
&  2.44 READ
! โครงสรางไวยากรณ
READ [SAVE]
P วัตถุประสงค
ค ําสัง่นี ้ใชคูกับคํ าสั่ง GET เพื่อรับขอมูลไปเก็บในตัวแปร
: ตัวอยางที่ 2.54
        CLS
        X1 = "ABCD"
        X2 = 0
        @ 5,5 SAY "GET X1 : " GET X1
        @ 6,5 SAY "GET X2 : " GET X2
        READ

&  2.45 RECALL
! โครงสรางไวยากรณ
RECALL [<ชวงทีต่องการ>] [FOR <เงื่อนไข>] [WHILE <เงื่อนไข>]
P วัตถุประสงค
ค ําสัง่นี ้ยกเลิกการทํ าเครื่องหมายลบหนาเรคอรด
: ตัวอยางที่ 2.55
        CLS
        USE FILEA
        DELETE FOR FIELD2 >= 1000
        LIST FIELD1,FIELD2,FIELD3
        OPT = " "
        @ ROW()+1,10 SAY "ARE YOU SURE TO DELETE?" GET OPT
        READ
        IF OPT $ [NN]
           RECALL ALL
        ELSE
           PACK
        ENDIF
        LIST FIELD1,FIELD2,FIELD3

&  2.46 REINDEX
! โครงสรางไวยากรณ
REINDEX
P วัตถุประสงค
ค ําสัง่นี ้สั่งใหจัดทํ าแฟมดรรชนีที่กํ าลังเปดอยูใหม
: ตัวอยางที่ 2.56
        USE FILEA INDEX IFIELD1,IFIELD2,IFIELD3
        REINDEX
        SET ORDER TO 2
        LIST FIELD1,FIELD2,FIELD3

&  2.47 RELEASE
! โครงสรางไวยากรณ
RELEASE <รายการชื่อตัวแปร>
RELEASE ALL [LIKE | EXCEPT <ชือ่ตัวแปร>]
P วัตถุประสงค
ค ําสั่งนี ้ยกเลิกตัวแปรแบบ PRIVATE และ PUBLIC
แตไมยกเลิกตัวแปรแบบ LOCAL และ STATIC
: ตัวอยางที่ 2.57
        LOCAL Z
        USE FILEA
        PRIVATE X
        PUBLIC Y
        X = 5
        Z = 6
        RELEASE X       // ยกเลิกเฉพาะตัวแปร X
        RELEASE ALL     // ยกเลกิการจองเนื้อที่หนวยความจ ําให Y
        ? Z
        LIST FIELD1,FIELD2,FIELD3

&  2.48 RENAME
! โครงสรางไวยากรณ
RENAME <ชือ่แฟมเดิม> TO <ชื่อแฟมใหม>

P วัตถุประสงค
ค ําสั่งนี้ เปลี่ยนชื่อแฟม
: ตัวอยางที่ 2.58
        RENAME TEST.TXT TO TEST1.TXT
        DIR TEST1.TXT
        X = "TEST.TXT"
        X1 = "TEST1.TXT"
        RENAME &X1 TO &X
        DIR TEST.TXT
        RENAME (X) TO (X1)
        DIR TEST1.TXT
        RENAME TEST1.TXT TO TEST.TXT

&  2.49 REPLACE
! โครงสรางไวยากรณ
REPLACE <ชือ่ฟลด> WITH <ชื่อตัวแปร>
     [,<ชือ่ฟลดที ่2> WITH <ชื่อตัวแปรที่ 2>...]
     [<ชวงทีต่องการ>] [FOR <เงื่อนไข>] [WHILE <เงื่อนไข>]
P วัตถุประสงค
ค ําสั่งนี้ แทนคาในฟลด
: ตัวอยางที่ 2.59
        CLS
        COPY FILE FILEA.DBF TO FILEC.DBF
        USE FILEA
        REPLACE FIELD3 WITH 0 ALL
        REPLACE FIELD1 WITH "101" FOR FIELD2 >= 1000
        ? "================="
        LIST FIELD1,FIELD2,FIELD3
        APPEND BLANK
        REPLACE FIELD1 WITH "105",;
                FIELD2 WITH 1000 , FIELD3 WITH 250
        ? "================="
        LIST FIELD1,FIELD2,FIELD3
        CLOSE

        COPY FILE FILEC.DBF TO FILEA.DBF
        ? "================="
        USE FILEA
        LIST FIELD1,FIELD2,FIELD3

&  2.50 RESTORE
! โครงสรางไวยากรณ
RESTORE FROM <ชือ่แฟมเก็บตัวแปร> [ADDITIVE]
P วัตถุประสงค
ค ําสัง่นี ้เรียกตัวแปรที่เคยเก็บไวในแฟมมาใชอีกครั้ง
: ตัวอยางที่ 2.60
        X1 := 5
        X2 := 10
        SAVE ALL TO XMEM
        CLEAR ALL
        RESTORE FROM XMEM
        ? X1,X2

&  2.51 RESTORE SCREEN
! โครงสรางไวยากรณ
RESTORE SCREEN [FROM <ตวัแปรเก็บจอภาพ>]
P วัตถุประสงค
ค ําสัง่นี ้เรียกรูปแบบของจอภาพที่เคยจัดเก็บไวมาแสดงบนจออีกครั้ง
: ตัวอยางที่ 2.61
        CLS
        SET COLOR TO "B+/W"
        ? "ABC"
        SAVE SCREEN  //  ถาไมมีการจัดเก็บรูปแบบของจอภาพหลายครั้ง
        SETCOLOR("GR+/B")
        ? "DEF"
        INKEY(0)
        RESTORE SCREEN
        INKEY(0)

: ตัวอยางที่ 2.62
        CLS
        SETCOLOR("W/B")
        @ 0,0 TO 24,79
        @ 2,1 TO 2,78
        @ 1,10 SAY PADC ("COMPUTER HEADING "+DTOC(DATE()),60)
        INKEY(0) ;  SAVE SCREEN TO SCR1
        @ 5,10 TO 10,20
        @ 7,15 TO 12,25
        INKEY(0) ;  SAVE SCREEN TO SCR2
        CLS
        OPT = " "
        @ 5,10 SAY "PRESS 1 OR 2 TO CHOOSE SCREEN : " GET OPT
        READ
        DO CASE
           CASE OPT = "1" ; RESTORE SCREEN FROM SCR1
           CASE OPT = "2" ; RESTORE SCREEN FROM SCR2
        ENDCASE
        INKEY(0)

&  2.52 RUN
! โครงสรางไวยากรณ
RUN | ! <ค ําสั่งภายนอก>
P วัตถุประสงค
ค ําสั่งนี้ สั่งประมวลผลโปรแกรมจากภายนอก
: ตัวอยางที่ 2.63
        CLS
        ? "IF YOU WANT TO RETURN TO APPLICATION,TYPE EXIT."
        ? "==============================================="
        INKEY(3)
        RUN COMMAND     // จะออกจากโปรแกรมไปอยูในสวน DOS SHELL
        ? "WELCOME TO APPLICATION"
        ? "======================"

: ตัวอยางที่ 2.64
        RUN DIR
        RUN TYPE X.PRG

&  2.53 SAVE
! โครงสรางไวยากรณ
SAVE TO <ชือ่แฟมเกบ็ตัวแปร> [ALL [LIKE | EXCEPT <ชื่อแฟม>]]
P วัตถุประสงค
ค ําสัง่นี ้จัดเก็บตัวแปรลงแฟม เพื่อน ํากลับมาใชในภายหลังได
: ตัวอยางที่ 2.65
        X01 := "A"
        X02 := 5
        Y01 := 6
        Y02 := "B"
        SAVE ALL TO VARALL
        CLEAR ALL
        RESTORE FROM VARALL
        ? X01,X02,Y01,Y02
: ตัวอยางที่ 2.66
        CLS
        ? "PASSWORD SYSTEM"
        ? "---------------"
        IF FILE("PASSF.MEM")
           RESTORE FROM PASSF
        ELSE
           ? "INVALID ON PASSWORD LIST"
           USER = "    "
           PASS = "    "
           @ 5,5 SAY "    NEW USER : " GET USER
           @ 6,5 SAY "NEW PASSWORD : " GET PASS
           READ
           PASS = STR(ASC(LEFT(PASS,1)))+RIGHT(PASS,3)
        ENDIF
        _USER := "    "

        _PASS := "    "
        WHILE .T.
           @ 8,5 SAY "    YOUR USER : " GET _USER
           @ 9,5 SAY "YOUR PASSWORD : " GET _PASS
           READ
           IF _USER = USER .AND. ;
              _PASS = CHR(VAL(LEFT(PASS,LEN(PASS)-3)))+RIGHT(PASS,3)
              SAVE ALL TO PASSF
              EXIT
           ENDIF
        END

&  2.54 SAVE SCREEN
! โครงสรางไวยากรณ
SAVE SCREEN [TO <ตวัแปรเก็บจอภาพ>]
P วัตถุประสงค
ค ําสั่งนี้ จัดเก็บรูปแบบของจอภาพ
: ตัวอยางที่ 2.67
        SETCOLOR("GR+/B")
        CLS
        @ 5,10 TO 15,20
        @ 15,5 TO 21,21 DOUBLE
        SAVE SCREEN TO SCR1
        SETCOLOR("W/BR+")
        @ 6,7,10,20 BOX '123456789'
        ? '-----------------------'
        SAVE SCREEN TO SCR2
        SETCOLOR("BG+/R+")
        @ 7,10,9,15 BOX '.'
        @ 9,21 CLEAR TO 18,24
        SAVE SCREEN TO SCR3
        SAVE ALL TO SCRF
        CLEAR ALL
        RESTORE FROM SCRF

        RESTORE SCREEN FROM SCR1   ; INKEY(3)
        RESTORE SCREEN FROM SCR2   ; INKEY(3)
        RESTORE SCREEN FROM SCR3   ; INKEY(3)

&  2.55 SEEK
! โครงสรางไวยากรณ
SEEK <EXPSEARCH>
P วัตถุประสงค
ค ําสัง่นี ้การคนหาจากฟลดที่ถูกกํ าหนดใหจัดเรียงในแฟมดรรชนี
: ตัวอยางที่ 2.68
        USE FILEA  INDEX IFIELD1
        SEEK "101"
        IF FOUND()
           ? FIELD1,FIELD2,FIELD3
        ENDIF
: ตัวอยางที่ 2.69
        USE FILEA  INDEX IFIELD1
        X = "101"
        @ ROW()+1,10 SAY "WHAT ID : " GET X
        READ
        SEEK X
        IF FOUND()
           ? FIELD1,FIELD2,FIELD3
           ? RECNO()
        ENDIF

&  2.56 SELECT
! โครงสรางไวยากรณ
SELECT <เลขประจ ําพื้นที>่ | <สมนาม>
P วัตถุประสงค
ค ําสั่งนี้ เลือกพื้นที่ท ํางาน
: ตัวอยางที่ 2.70
        USE FILEA
        USE FILEB
        LIST FIELD1,FIELD2,FIELD3   // แสดงขอมูลจากแฟม FILEB

        SELE 1
        LIST FIELD1,FIELD2,FIELD3   // แสดงขอมูลจากแฟม FILEA
        SELE FILEB
        ? RECCOUNT()                // แสดงจ ํานวนเรคอรดของ FILEB

&  2.57 SET ALTERNATE
! โครงสรางไวยากรณ
SET ALTERNATE TO [<ชือ่แฟม> [ADDITIVE]]
SET ALTERNATE ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชสรางสรางแฟม เพื่อรองรับการเขียนลงแฟมดวยคํ าสั่ง ? หรือ LIST
การใชค ําสั่ง ? หรือ LIST จะเห็นผลบนจอภาพปกต ิและเขียนลงแฟมพรอม ๆ กัน
หากระบชุื่อแฟมแตไมมีนามสกุล โปรแกรมจะตั้งนามสกุลชื่อ TXT ใหอัตโนมัติ
โดย ON หมายถึง เปดแฟมเพื่อรับรองรับการเขียนดวยคํ าสั่ง ?
และ OFF หมายถึง ไมเปดแฟมที่ระบุไวนั้นขึ้นมา
: ตัวอยางที่ 2.71
        SET ALTERNATE TO XX ADDITIVE   // TO MAKE XX.TXT
        SET ALTERNATE ON
        ? 'ABC'
        ? 'DEF'   ;        ? 'GHI'
        @ 5,5 SAY 'ABC' // NO EFFECT ON XX.TXT
        SET ALTERNATE TO XX.XXX       // TO MAKE XX.XXX
        ? 'ABC'
        CLOSE ALTERNATE

&  2.58 SET BELL
! โครงสรางไวยากรณ
SET BELL ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชควบคุมการเปลงเสียง เมื่อปอนขอมูลจนลนชองรับขอมูล
ในขณะทีใ่ชคํ าสั่ง GET แตไมมีผลตอฟงกชัน TONE() และ CHR(7)
โดย ON หมายถึง ใหเกิดเสียงเมื่อปอนขอมูลจนเกินชองที่เตรียมไว
และ OFF หมายถึง ไมใหเกิดเสียงเตือน

: ตัวอยางที่ 2.72
        SET BELL OFF
        X = 5
        @ ROW(),10 SAY "GET X1 " GET X  // ไมมีเสียงเมื่อปอนขอมูลมากเกิน
        READ
        TONE(200,20) ;  ?? CHR(7)
        SET BELL ON
        X = 5
        @ ROW(),60 SAY "GET X2 " GET X  // มเีสยีงเมื่อปอนขอมูลมากเกิน
        READ
        TONE(200,20) ;  ?? CHR(7)

&  2.59 SET CENTURY
! โครงสรางไวยากรณ
SET CENTURY ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสั่งนี้ ท ําใหการแสดงผลของป ค.ศ. เปน 2 หลัก หรือ 4 หลัก
โดย ON หมายถึง ใหแสดงป ค.ศ. เปน 4 หลัก เชน 2002
และ OFF หมายถึง ใหแสดงป ค.ศ. เปน 2 หลัก เชน 02
: ตัวอยางที่ 2.73
        SET CENTURY ON
        ? DATE()        //   08/23/2002
        SET CENTURY OFF
        ? DATE()        //   08/23/02

&  2.60 SET COLOR
! โครงสรางไวยากรณ
SET COLOR | COLOUR TO
    [[<สขีองตวัปกต>ิ] [,<สีของการรับคา>] [,<สีของเสน>] [,<สีของพื้น>]
     [<สขีองสวนที่ไมไดรับคา>]] | (<ระบุส>ี)
P วัตถุประสงค
ค ําสัง่นี ้จะกํ าหนดสีใหกับสวนตาง ๆ ของจอภาพ
โดย สขีองตัวปกติ หมายถึง สีของตัวอักษรที่พิมพดวยคํ าสั่ง ? หรือ SAY
    สขีองการรับคา หมายถึง สีที่ปรากฏในสวนที่ก ําลังรอรับคาดวยคํ าสั่ง GET

    สขีองเสน หมายถึง สีของเสนขอบ ซึ่งเปนเสนนอกสุด
    สีของพื้น หมายถึง
    สขีองสวนที่ไมไดรับคา หมายถึง สีที่ปรากฏในสวนที่ใชคํ าสั่ง GET แตยังไมรอรับคา

ชื่อส ี                ตวัอักษรยอ    ตัวเลข  ส ําหรับจอสีเดียว
ด ํา (BLACK)              N,SPACE         0       BLACK
นํ ้าเงิน (BLUE)          B               1       UNDERLINE
เขียว (GREEN)           G               2       WHITE
คราม (CYAN)             BG              3       WHITE
แดง (RED)               R               4       WHITE
มวง (MAGENTA)          RB              5       WHITE
นํ ้าตาล (BROWN)          GR              6       WHITE
ขาว (WHITE)             W               7       WHITE
เทา (GRAY)              N+              8       BLACK
นํ ้าเงินสวาง (BRIGHT BLUE) B+           9       BRIGHT UNDERLINE
เขยีวสวาง (BRIGHT GREEN) G+           10       BRIGHT WHITE
ครามสวาง (BRIGHT CYAN)  BG+           11       BRIGHT WHITE
แดงสวาง (BRIGHT RED)      R+             12       BRIGHT WHITE
มวงสวาง (BRIGHT MAGENTA) RB+          13       BRIGHT WHITE
เหลอืง (YELLOW)         GR+            14       BRIGHT WHITE
ขาวสวาง (BRIGHT WHITE)  W+            15       BRIGHT WHITE

: ตัวอยางที่ 2.74
        SET COLOR TO "B/G,BG/R,RB,,N+/GR"
        CLS
        @ 5,5 TO 10,15 DOUBLE     //  B/G
        ? "TEST COLOR"            //  B/G
        X = 5
        Y = 15
        @ 15,5 SAY "TEST COLOR X " GET X  // BG/R
        @ 16,5 SAY "TEST COLOR Y " GET Y  // N+/GR
        READ
        WAIT TO X

&  2.61 SET CONFIRM
! โครงสรางไวยากรณ
SET CONFIRM ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชรวมกับค ําสั่ง GET เพราะจะควบคุมการออกจาก GET อยางไมถูกตอง
จนกวาจะกดปุมเอ็นเทอร (ENTER) เพื่อขามไปยังบรรทัดตอไป
โดย ON หมายถึง ถาปอนขอมูลจนลนชองแลวจะไมหลุดไปยังบรรทัดตอไป
และ OFF หมายถึง ถาปอนขอมูลจนลน จะขามไปบรรทัดตอไปอัตโนมัติ
: ตัวอยางที่ 2.75
        SET CONFIRM ON
        CLS
        X = 5
        @ 15,5 SAY "TEST CONFIRM " GET X
        READ

&  2.62 SET CONSOLE
! โครงสรางไวยากรณ
SET CONSOLE ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชระบุใหการแสดงผลทางจอภาพเกิดขึ้นหรือไม
โดย ON หมายถึง ใหแสดงผลทางจอภาพ
และ OFF หมายถึง ไมใหแสดงผลทางจอภาพ
: ตัวอยางที่ 2.76
        SET CONSOLE ON
        USE FILEA
        LIST FIELD1 TO PRINTER  // ออกทั้งจอภาพ และเครื่องพิมพ
        SET CONSOLE OFF
        LIST FIELD1 TO PRINTER // ออกเฉพาะเครื่องพิมพ

&  2.63 SET CURSOR
! โครงสรางไวยากรณ
SET CURSOR ON | OFF | <ตรรกนิพจน>

P วัตถุประสงค
ค ําสัง่นี ้ท ําใหตัวกระพริบ(CURSOR) ปรากฏ หรือไมปรากฏบนจอภาพ
โดย ON หมายถึง ปรากฏตัวกระพริบ(CURSOR) บนจอภาพ
และ OFF หมายถึง ไมปรากฏตัวกระพริบ(CURSOR) บนจอภาพ
: ตัวอยางที่ 2.77
        SET CURSOR OFF
        WAIT
        INKEY(5)
        X = 5
        @ ROW(),COL() GET X
        READ

&  2.64 SET DATE
! โครงสรางไวยากรณ
SET DATE FORMAT [TO] <รปูแบบวันที ่ที่ตองการ>
SET DATE [TO] AMERICAN | ANSI | BRITISH | FRENCH |
              GERMAN | ITALIAN | JAPAN | USA
P วัตถุประสงค
ค ําสัง่นี ้ใชเลือกรูปแบบของวันที่ ที่จะน ําไปใชในโปรแกรม

     คาติดตั้ง      รูปแบบที่ได
     AMERICAN        MM/DD/YY
     ANSI            YY.MM.DD
     BRITISH         DD/MM/YY
     FRENCH          DD/MM/YY
     GERMAN          DD.MM.YY
     ITALIAN         DD-MM-YY
     JAPAN           YY/MM/DD
     USA             MM-DD-YY

: ตัวอยางที่ 2.78
        SET DATE FORMAT TO "DD/MM/YYYY"
        ? DATE()                //   28/08/2002
        SET CENTURY OFF
        SET DATE TO GERMAN

        ? DATE()                //   28.08.02
        SET DATE TO ANSI
        ? DATE()                //   02.08.28
        SET CENTURY ON
        SET DATE TO GERMAN
        ? DATE()                //   28.08.2002
        SET DATE TO ANSI
        ? DATE()                //   2002.08.28

&  2.65 SET DECIMALS
! โครงสรางไวยากรณ
SET DECIMALS TO [<จ ํานวนทศนิยม>]
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดจํ านวนทศนิยม แตผลการคํ านวณจะไดทศนิยมไมเกิน 15 หลัก
: ตัวอยางที่ 2.79
        SET DECIMALS TO 2
        ? 22/7              //  3.14
        SET DECIMALS TO 20
        ? 22/7              //  3.14285714285714300000
        SET DECIMALS TO 30
        ? 22/7             //  3.142857142857143000000000000000
        SET DECIMALS TO
        ? 22/7             //  3

&  2.66 SET DEFAULT
! โครงสรางไวยากรณ
SET DEFAULT TO [<ขอความระบุไดรฟและไดเรกทรอรี่ที่เก็บขอมูล>]
P วัตถุประสงค
ค ําสัง่นี ้ใชกํ าหนดไดเรกทรอรี่ที่เก็บขอมูล แตไมเปลี่ยนไดเรกทรอรี่
เหมอืนค ําสั่งเปลี่ยนไดเรกทรอรี่ในดอส (CD : CHANGE DIRECTORY)
: ตัวอยางที่ 2.80
        SET DEFAULT TO C:\LANGUAGE\CLIPPER5
        ! DIR
        DIR                    // คนแฟมในไดเรกทรอรี่นี้

        SET DEFAULT TO C:      // เปลี่ยนมาใชไดรฟ C
        SET DEFAULT TO ..      // ออกไปนอกไดเรกทรอรี่ปจจุบัน 1 ระดับ
        ! DIR                  // ยงัอยูที ่LANGUAGE\CLIPPER5
        DIR                    // ไมพบแฟมใน LANGUAGE
        SET PATH TO C:\;C:\WINDOWS;C:\DOS;C:\LANGUAGE\CLIPPER5
        SET DEFAULT TO \       // กลับไปที่ไดเรกทรอรี่ราก
        ! DIR                  // ยงัอยูที ่LANGUAGE\CLIPPER5
        DIR                    // ไมพบแฟมใน C:\

&  2.67 SET DELETED
! โครงสรางไวยากรณ
SET DELETED ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ท ําใหเรคอรดที่ถูกทํ าเครื่องหมายลบ ถูกมองวาไมอยูจริง ๆ
ซึง่ใหผลคลายค ําสั่ง SET FILTER TO !DELETE()
โดย ON หมายถึง มองวาเรคอรดที่ถูกทํ าเครื่องหมายลบ ถูกลบไปแลวจริง ๆ
หรอื OFF หมายถึง มองวาเรคอรดที่ถูกทํ าเครื่องหมายลบ ยังมีอยูในแฟม
: ตัวอยางที่ 2.81
        USE FILEA                  // แฟมนี้มีขอมูล 4 เรคอรด
        RECALL ALL
        X = 0
        DELETE RECORD 2
        DELETE RECORD 4
        LIST FIELD1,FIELD2,FIELD3  // เห็น * หนาเรคอรดที่ 2 และ 4
        COUNT TO X ; ? X           // 4
        SET DELETED ON
        COUNT TO X ; ? X           // 2
        SET DELETED OFF            // ตองเซตให OFF จึงจะสั่ง RECALL ได
        RECALL ALL
        LIST FIELD1,FIELD2,FIELD3  // ไมมีเรคอรดที่ถูกท ําเครื่องหมายลบ
        COUNT TO X ; ? X           // 4
        DELETE RECORD 1
        SET FILTER TO !DELETED()
        COUNT TO X ; ? X           // 3

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


52                                             การเขียนโปรแกรมสํ าหรับฐานขอมูลดวย Clipper                                                         

&  2.68     SET DELIMITERS
! โครงสรางไวยากรณ
SET DELIMITERS ON | OFF | <ตรรกนิพจน>
SET DELIMITERS TO [<ตวัอกัษร> | DEFAULT]
P วัตถุประสงค
ค ําสัง่นี ้ใชกํ าหนดตัวอักษรที่จะอยูขางหนาและขางหลัง
ของตวัแปรที่รับคาดวยค ําสั่ง GET ซึ่งทํ าใหผูใชเห็นฟลดไดชัดเจนขึ้น
ตวัอกัษรจะใช 2 ตัวอักษร คืออยูขางหนา และขางหลัง อยางละตัว
โดย ON หมายถึง ใหใชตัวอักษรคลุมฟลดในขณะใชคํ าสั่ง GET
และ OFF หมายถึง ไมใหมีตัวอักษรคลุมฟลด
: ตัวอยางที่ 2.82
        X := Y := SPACE(5)
        CLS
        SET DELIMITERS ON
        SET DELIMITERS TO ": "   // มีเฉพาะหนาฟลด
        @ 5,5 SAY "GET X" GET X
        SET DELIMITERS TO "[ ]"   // มี [ อยูหนาฟลด และ ] อยูหลังฟลด
        @ 6,5 SAY "GET Y" GET Y
        READ

&  2.69 SET DEVICE
! โครงสรางไวยากรณ
SET DEVICE TO SCREEN | PRINTER
P วัตถุประสงค
ค ําสัง่นี ้ใชระบุอุปกรณที่จะแสดงผล และมักใชคูกับคํ าสั่ง SAY
: ตัวอยางที่ 2.83
        SET DEVICE TO PRINTER
        @ 5,5 SAY "WOW"
        @ 6,5 SAY "WOW"
        @ 1,5 SAY "WOW"         // จะไปพิมพที่บรรทัดที ่1 ของหนาตอไป
        EJECT
: ตัวอยางที่ 2.84
        SET PRINTER TO "OUT.TXT"
        SET DEVICE TO PRINTER

        ? "==="              // ใหผลทางจอภาพเทานั้น
        @ 5,5 SAY "WOW"      // ผลลัพธทั้ง 3 บรรทัดปรากฏในแฟม OUT.TXT
        @ 6,5 SAY "WOW"      // แตไมปรากฏบนจอภาพ
        @ 8,5 SAY "WOW"

&  2.70 SET EPOCH
! โครงสรางไวยากรณ
SET EPOCH TO <ป>
P วัตถุประสงค
ค ําสัง่นี ้ท ําใหโปรแกรมฉลาดขึ้นในเรื่องการมองป จากการที่โปรแกรมมักรับป
เปนเลข 2 หลัก แลวเติมศตวรรษให หากตองการป 2005 ควรเติมเพียง 05
หากใชคาปริยาย จะไดปเปน 1905 ทันท ีซึ่งคาปริยายกํ าหนดให EPOCH เปน
1900 ชวงปทีเ่ปนไปไดคือ 1900 ถึง 1999 ดังนั้นเมื่อเติม 12 จึงเปน 1912
   ถาตองการเติม 01 แลวใหเปน 2001 ควรเซต EPOCH เปน 1960
ท ําใหปที่เปนไปไดเมื่อเติมเลข 2 หลักคือ 1960 ถึง 2059
: ตัวอยางที่ 2.85
        SET DATE FORMAT TO "DD/MM/YYYY"
        ? CTOD("15/06/01")      //  15/06/1901
        SET EPOCH TO 1960       //  ชวงของปคือ 1960 - 2059
        ? CTOD("15/06/01")      //  15/06/2001
        SET EPOCH TO 1995       //  ชวงของปคือ 1995 - 2094
        ? CTOD("15/06/92")      //  15/06/2092

&  2.71 SET ESCAPE
! โครงสรางไวยากรณ
SET ESCAPE ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชอนุญาติหรือจ ํากัด การใชปุมเอสเคป (ESC KEY) กับคํ าสั่ง READ
โดย ON หมายถึง ใหออกจากคํ าสั่ง READ ดวยการกดปุมเอสเคป (ESC KEY)
หรอื OFF หมายถึง ไมใหออกจากคํ าสั่ง READ ดวยการกดปุมเอสเคป (ESC KEY)
: ตัวอยางที่ 2.86
        SET ESCAPE OFF
        X := Y:= Z:= 5
        CLS

        @ 5,5 GET X     // ไมสามารถใชปุมเอสเคปตรงนี้ได
        @ 6,5 GET Y
        @ 7,5 GET Z
        READ
        SET ESCAPE ON
        @ 8,5 GET X     // ถากด ESC ตรงนี ้จะหลุดจากการรับ Y ดวย
        @ 9,5 GET Y
        READ

&  2.72 SET EXACT
! โครงสรางไวยากรณ
SET EXACT ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชคุมการเปรียบเทียบในเรื่องความเทากันของนิพจน 2 ตัว
โดย ON หมายถึง เปนจริงตอเมื่อทั้งนิพจนแรกและหลังเหมือนกันทุกประการ
หรอื OFF หมายถึง เปนจริงตอเมื่อทั้งหมดของนิพจนหลังเทากับนิพจนแรก
: ตัวอยางที่ 2.87
        SET EXACT OFF
        ? "123" = "123"         // .T.
        ? "123 " = "123"        // .T.
        ? "123" = "12"          // .T.
        ? "12" = ""             // .T.
        ? "123" = "123 "        // .F.
        ? "12" = "123"          // .F.
        ? "" = "123"            // .F.
        SET EXACT ON
        ? "123" = "123"         // .T.
        ? "123 " = "123"        // .T.
        ? "123" = "12"          // .F.
        ? "12" = ""             // .F.
        ? "123" = "123 "        // .T.
        ? "12" = "123"          // .F.
        ? "" = "123"            // .F.

&  2.73 SET EXCLUSIVE
! โครงสรางไวยากรณ
SET EXCLUSIVE ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชควบคุมแฟมโดยตรง
โดย ON หมายถึง แฟมไมสามารถถูกแบงใหใครใชไดในขณะที่โปรแกรมนี้ใชงาน
และ OFF หมายถึง แฟมสามารถถูกแบงใหผูใชคนอื่น ๆ ใชรวมกันได
: ตัวอยางที่ 2.88
        USE FILEA
        SET EXCLUSIVE ON
        LIST FIELD1,FIELD2,FIELD3

&  2.74 SET FILTER
! โครงสรางไวยากรณ
SET FILTER TO [<เงือ่นไข>]
P วัตถุประสงค
ค ําสัง่นี ้ใชจ ํากัดชวงขอมูลที่นํ ามาใช ใหมีขอบเขตที่แคบลงมา เชนตองการ
ขอมลูของคนที่เงินเดือน มากกวา 5000 บาทเปนตน
: ตัวอยางที่ 2.89
        USE FILEA
        SET FILTER TO FIELD2 > 5000
        LIST FIELD1,FIELD2,FIELD3
        SET FILTER TO           // ยกเลิกการใชตัวกรอง
        LIST FIELD1,FIELD2,FIELD3
        SET FILTER TO DELETE()  // เลอืกเฉพาะขอมูลที่ถูกท ําเครื่องหมายลบ
        LIST FIELD1,FIELD2,FIELD3

&  2.75 SET FIXED
! โครงสรางไวยากรณ
SET FIXED ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชควบคุมใหการก ําหนดตํ าแหนงทศนิยมทํ างาน หรือใชคาโดยปริยายแทน
โดย ON หมายถึง ใหใชจํ านวนตํ าแหนงทศนิยมตามคํ าสั่ง SET DECIMALS
หรอื OFF หมายถึง ใหจํ านวนตํ าแหนงทศนิยมเปนไปตามคาปริยาย

: ตัวอยางที่ 2.90
        SET FIXED ON
        SET DECIMALS TO 5
        X = 5
        Y = 2
        Z = X / Y
        ? Z                // 2.50000

&  2.76 SET FORMAT
! โครงสรางไวยากรณ
SET FORMAT TO [<ชือ่โปรแกรมยอย>[.<สวนขยาย>]]
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดโปรแกรมยอยที่จะท ํางาน มักใชรวมกับคํ าสั่ง READ
ท ําใหสามารถน ําไปเลือกรูปแบบที่จะใชรับขอมูลได
ในบางกรณีอาจเห็นผลของค ําสั่ง DO ใหผลเหมือนค ําสั่ง SET FORMAT
แตตามตวัอยางที่ขางลางไมสามารถใชค ําสั่ง DO ได เพราะคํ าสั่ง DO จะไม
สงรปูแบบคืนมาให READ ท ํางานในลูปไดอีก ท ําใหโปรแกรมคาง (HANG)
: ตัวอยางที่ 2.91
        CLS
        FRM = 1
        X = 5
        Y = 10
        CLS
        SET DELIMITERS TO "[]"
        SET DELIMITERS ON
        @ 4,5 SAY "CHOOSE FORM 1 TO 3 " GET FRM PICT "9"
        READ                         // เลือกรูปแบบที่จะปอนขอมูลได
        DO CASE
           CASE FRM = 1 ;  SET FORMAT TO TESTPROC1
           CASE FRM = 2 ;  SET FORMAT TO TESTPROC2
           CASE FRM = 3 ;  SET FORMAT TO TESTPROC3
        ENDCASE
        WHILE LASTKEY() != 27
           READ

           ? X,Y,X+Y,X-Y,X*Y,X/Y
        END
        PROCEDURE TESTPROC1
           @ 5,5  SAY "GET X" GET X
           @ 6,5  SAY "GET Y" GET Y
        RETURN
        PROCEDURE TESTPROC2
           @ 5,5  SAY "GET X" GET X
           @ 5,55 SAY "GET Y" GET Y
        RETURN
        PROCEDURE TESTPROC3
           @ 5,4 TO 8,35 DOUBLE
           @ 6,5  SAY "GET X" GET X
           @ 7,5  SAY "GET Y" GET Y
        RETURN

&  2.77 SET FUNCTION
! โครงสรางไวยากรณ
SET FUNCTION <เลขทีปุ่มฟงกชัน> TO <ขอความ>
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดขอความประจํ าปุมฟงกชัน เมื่อกดปุมฟงกชันแลวจะนํ า
ขอความไปปอนในฟลดที่ก ําลังรอรับคาอยูได ท ําใหสะดวกในการใสขอความ
ทีม่กัตองใชบอย ๆ เชนชื่อบริษัท หรือชื่อผูจัดการเปนตน

เลขประจ ําปุมฟงกชัน   ชือ่ปุมฟงกชัน
1-10                    F1         - F10
11-20                   SHIFT-F1 - SHIFT-F10
21-30                   CTRL-F1  - CTRL-F10
31-40                   ALT-F1   - ALT-F10

: ตัวอยางที่ 2.92
        SET KEY -2 TO EXITPROC        // -2 หมายถึงปุม F3
        SET FUNCTION 5 TO "YONOK COLLEGE"
        SET FUNCTION 6 TO "DR.NIRUND JIVASANTIKARN"
        CLS
        X = SPACE(25) // กด F5 จะสงคํ าวา YONOK COLLEGE ลงใน GET

        CNT = 0       // ระบคุ ําที่ใชบอย ๆ ในปุมฟงกชันเพื่อความสะดวก
        @ 1,1 SAY "F3:EXIT  F5:YNK   F6:NIRUND"
        DO WHILE .T.
           CNT++
           @ CNT+1,10 SAY STR(CNT,3)+" TEST FUNCTION :" GET X
           READ
        ENDDO
        PROCEDURE EXITPROC
            QUIT
        RETURN

&  2.78 SET INDEX
! โครงสรางไวยากรณ
SET INDEX TO <รายการชื่อแฟมดรรชน>ี
P วัตถุประสงค
ค ําสัง่นี ้ใชเปดแฟมดรรชนี ในกรณีที่เปดแฟมแลวไมไดเปดแฟมดรรชนีพรอมกัน
เพราะปกติโปรแกรมสามารถเปดแฟมพรอมเปดแฟมดรรชนีไดพรอมกัน
ค ําสัง่นีจ้งึใชเปดแฟมดรรชนีใหม หรือยกเลิกการเปดแฟมดรรชนีได
: ตัวอยางที่ 2.93
        USE FILEA INDEX IFIELD1,IFIELD2
        LIST FIELD1,FIELD2,FIELD3
        SET ORDER TO 2
        LIST FIELD1,FIELD2,FIELD3
: ตัวอยางที่ 2.94
        USE FILEA
        LIST FIELD1,FIELD2,FIELD3
        SET INDEX TO IFIELD1,IFIELD2
        LIST FIELD1,FIELD2,FIELD3
        SET INDEX TO
        LIST FIELD1,FIELD2,FIELD3

&  2.79 SET INTENSITY
! โครงสรางไวยากรณ
SET INTENSITY ON | OFF | <ตรรกนิพจน>

P วัตถุประสงค
ค ําสัง่นี ้ใชคุมการใชสีของคํ าสั่ง GET ที่จะใชสีของการรับคา หรือสีมาตรฐาน
โดย ON หมายถึง ใหคํ าสั่ง GET ใชสีของการรับคา หรือสวนที่ยังไมไดรับคา
และ OFF หมายถึง ใหคํ าสั่ง GET ใชสีมารตฐานทุกกรณี
: ตัวอยางที่ 2.95
        SET COLOR TO "B/G,BG/R,RB,,N+/GR"
        CLS
        X = 5
        Y = 15
        @ 13,5 SAY "TEST COLOR X " GET X  // BG/R เมื่อรอรับคา
        @ 14,5 SAY "TEST COLOR Y " GET Y  // N+/GR
        READ
        SET INTENSITY OFF  // ยกเลิกการเซตสีในสวนของค ําสั่ง GET
        CLS               // โดยใหสวนของค ําสั่ง GET มาใชสีมาตรฐาน
        @ 15,5 SAY "TEST COLOR X " GET X // B/G
        @ 16,5 SAY "TEST COLOR Y " GET Y // B/G
        READ

&  2.80 SET KEY
! โครงสรางไวยากรณ
SET KEY <รหสัการกดแปนพิมพ> TO [<ชื่อโปรแกรมยอย>]
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดโปรแกรมยอยประจ ําปุมฟงกชัน นั่นหมายถึงเมื่อ
กดปุมฟงกชันจะท ําใหโปรแกรมยอยที่ระบุไวท ํางานทันที
: ตัวอยางที่ 2.96
        SET KEY 28 TO HELPPROC        // 28 หมายถึงปุม F1
        SET KEY -1 TO CALPROC         // -1 หมายถึงปุม F2
        SET KEY -2 TO EXITPROC        // -2 หมายถึงปุม F3
        CLS
        X = SPACE(25)
        CNT = 0
        @ 1,1 SAY "F1:HELP"
        DO WHILE .T.
           CNT++

           @ CNT+1,10 SAY STR(CNT,3)+" TEST FUNCTION :" GET X
           READ
        ENDDO
        PROCEDURE HELPPROC
           X = ALERT("F1:HELP  F2:PUT CAL. VALUE  F3:EXIT",{"OK"})
           // เมือ่แสดงสวนชวยเหลือแลว ถากด  Enter  จะสง 1 กลับไป
           // แตถากดปุม ESC จะสงเลข 0 กลับไปให X
        RETURN
        PROCEDURE EXITPROC
           QUIT
        RETURN
        PROCEDURE CALPROC
           IF X < 5 ; X = CNT * 100  ; ENDIF
           IF X = 5 ; X = CNT * 500  ; ENDIF
           IF X > 5 ; X = CNT * 1000 ; ENDIF
        RETURN

&  2.81 SET MARGIN
! โครงสรางไวยากรณ
SET MARGIN TO [<เลขระบุกั้นหนา>]
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดตํ าแหนงกั้นหนาเพื่อพิมพรายงานออกเครื่องพิมพ
: ตัวอยางที่ 2.97
        USE FILEA
        SET MARGIN TO 8
        LIST FIELD1,FIELD2 TO PRINTER  // ไดกั้นหนาเปน 8
        SET MARGIN TO 5
        LIST SECOND(),FIELD1 TO PRINTER // ไดกั้นหนาเปน 5
        EJECT                           // ใหผลทั้งเครื่องพิมพและจอภาพ
        SET DEVICE TO SCREEN

&  2.82 SET MESSAGE
! โครงสรางไวยากรณ
SET MESSAGE TO [<แถว> [CENTER | CENTRE]]

P วัตถุประสงค
ค ําสัง่นี ้ใชระบุแถวที่จะแสดงขาวสารจากที่เคยระบุขาวสารในคํ าสั่ง PROMPT
: ตัวอยางที่ 2.98
        SET MESSAGE TO 22 CENTRE
        SET WRAP ON
        MSS = "แปลวา "
        @ 6,5 PROMPT "ANT"    MESSAGE MSS+"มด"
        @ 7,5 PROMPT "BAT"    MESSAGE MSS+"คางคาว"
        @ 8,5 PROMPT "CAT"    MESSAGE MSS+"แมว"
        @ 9,5 PROMPT "DOG"    MESSAGE MSS+"สุนัข"
        MENU TO OPT
        DO CASE
           CASE OPT = 0
                QUIT
           CASE OPT > 0
                ? "O.K."
        ENDCASE

&  2.83 SET ORDER
! โครงสรางไวยากรณ
SET ORDER TO [<เลขล ําดับแฟมดรรชนี>]
P วัตถุประสงค
ค ําสัง่นี ้ใชเลือกแฟมดรรชนีที่จะมาท ํางาน ซึ่งจัดเรียงตามฟลดที่กํ าหนด
ค ําสัง่นีจ้งึเปนค ําสั่งที่เลือกแฟมดรรชนีที่เคยเปดรอไวแลวเทานั้น
: ตัวอยางที่ 2.99
        USE FILEA
        INDEX ON FIELD1 TO IFIELD1
        INDEX ON FIELD2 TO IFIELD2
        SET INDEX TO IFIELD1,IFIELD2
        SET ORDER TO 2
        LIST FIELD1,FIELD2,FIELD3
        SET ORDER TO 1
        LIST FIELD1,FIELD2,FIELD3

&  2.84 SET PATH
! โครงสรางไวยากรณ
SET PATH TO [<รายการระบุเสนทาง>]
P วัตถุประสงค
ค ําสัง่นี ้ใชระบุเสนทางที่จะเรียกใชแฟม เมื่อไมพบในไดเรกทรอรี่ปจจุบัน
: ตัวอยางที่ 2.100
        ? FILE("COMMAND.COM")   // .F.
        SET PATH TO C:\;C:\DOS;C:\WINDOWS
        ? FILE("COMMAND.COM")   // .T.
        ? FILE("FORMAT.COM")    // .T.
        TYPE AUTOEXEC.BAT
        // แฟม AUTOEXEC.BAT อยูใน C:\ ที่หาพบ เพราะเซตเสนทางไว

&  2.85 SET PRINTER
! โครงสรางไวยากรณ
SET PRINTER ON | OFF | <ตรรกนิพจน>
SET PRINTER TO [<ชือ่อปุกรณ> | <ชื่อแฟม> [ADDITIVE]]
P วัตถุประสงค
ค ําสัง่นี ้ใชระบุปลายทางของผลลัพธเปนเครื่องพิมพ หรือแฟม
โดย ON หมายถึง ใหแสดงผลทางเครื่องพิมพ หรือน ําขอมูลออกแฟมได
และ OFF หมายถึง ไมอนุญาติใหใชเครื่องพิมพ ดวยคํ าสั่ง SET PRINTER TO
: ตัวอยางที่ 2.101
        SET PRINTER ON
        SET CONSOLE OFF
        USE FILEA
        WHILE .NOT. EOF()
           ? FIELD1,FIELD2,FIELD3  // ไมแสดงผลทางจอภาพ
           SKIP
        END
: ตัวอยางที่ 2.102
        SET PRINTER TO OUT.TXT
        SET PRINTER ON
        SET CONSOLE OFF

        USE FILEA
        WHILE .NOT. EOF()
           ? FIELD1,FIELD2,FIELD3  // น ําผลลัพธออกทางแฟมเทานั้น
           SKIP
        END

&  2.86 SET PROCEDURE
! โครงสรางไวยากรณ
SET PROCEDURE TO [<ชือ่โปรแกรมยอย>[.<สวนขยาย>]]
P วัตถุประสงค
ค ําสัง่นี ้ใชระบุโปรแกรมยอยจากภายนอกที่จะน ํามาแปล (COMPILE)
เพราะบางครั้งโปรแกรมตองเรียกใชโปรแกรมจากภายนอก แตไมไดระบุชื่อ
ของโปรแกรมไวแตแรก
: ตัวอยางที่ 2.103
        SET PROCEDURE TO Y.PRG
        PROCEXT = "Y"
        DO &PROCEXT
        // ถาโปรแกรมนี้ไมใชค ําสั่ง SET PROCEDURE จะประมวลผลไมผาน
        // เพราะไมไดเรียกโปรแกรม Y มาแปลพรอม ๆ กัน

&  2.87 SET RELATION
! โครงสรางไวยากรณ
SET RELATION TO [<นพิจน> | <ล ําดับของเรคอรด> INTO <สมนาม>]
    [, [TO] <นพิจนที ่2> | <ล ําดับของเรคอรดที่ 2> INTO <สมนามที่ 2>...]
    [ADDITIVE}
P วัตถุประสงค
ค ําสัง่นี ้ใชเชื่อมความสัมพันธของแฟมอัตโนมัต ิท ําใหเชื่อมกันและถูกเรียกใช
ไดอยางงายดาย โดยไมตองใชคํ าสั่งคนหาใหยุงยาก
: ตัวอยางที่ 2.104
        USE INVF INDEX IINV NEW
        USE ZIPF INDEX IZIP NEW
        USE CUSF
        SET RELATION TO CUSID INTO INVF , ZIPCODE INTO ZIPF
        LIST CUSID , INVF->INVNUM , INVF->DATE , ZIPF->PROVINCE

&  2.88 SET SCOREBOARD
! โครงสรางไวยากรณ
SET SCOREBOARD ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชอนุญาติใหแสดงขอผิดพลาดจากคํ าสั่ง READ หรือ MEMOEDIT ในบรรทัด 0
เชนการใสคาในค ําสั่ง READ เกิดชวงที่ระบุ จะแสดงขอผิดพลาดทางในบรรทัด 0
โดย ON หมายถึง อนุญาติใหใชบรรทัด 0 แสดงขอผิดพลาด
และ OFF หมายถึง ไมแสดงขอผิดพลาดในบรรทัด 0
: ตัวอยางที่ 2.105
        X = 5
        @ 5,5 GET X RANGE 0 , 99
        READ
        SET SCOREBOARD OFF        // ไมแสดงขอผิดพลาด
        @ 6,5 GET X RANGE 0 , 99  // ถาคาผิด จะใชคาปริยายแทนคาที่ผิด
        READ

&  2.89 SET SOFTSEEK
! โครงสรางไวยากรณ
SET SOFTSEEK ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้ใชควบคุมเกี่ยวกับการคนหาเรคอรด หากคนหาไมพบ
โดย ON หมายถึง ถาคนหาไมพบ ตัวชี้จะไปชี้ที่เรคอรดมีคาสูงกวา
และ OFF หมายถึง ถาคนหาไมพบ ตัวชี้จะไปชี้ที่ทายเรคอรดสุดทาย
: ตัวอยางที่ 2.106
        USE FILEA INDEX IFIELD2
        SEEK 1001
        IF .NOT. FOUND()
           ? FIELD2         // 0 ทายเรคอรดสุดทายไมมีคา
        ENDIF
        SET SOFTSEEK ON
        SEEK 1001
        IF .NOT. FOUND()
           ? FIELD2         // 6500 คาตอมาที่สูงกวา 1001
        ENDIF

&  2.90 SET TYPEAHEAD
! โครงสรางไวยากรณ
SET TYPEAHEAD TO <จ ํานวนเลขระบุขนาด>
P วัตถุประสงค
ค ําสัง่นี ้ใชก ําหนดขนาดหนวยความจํ าที่เก็บคาของแปนพิมพ เมื่อผูใชกด
แปนพมิพจนหนวยความจํ ารับไปประมวลผลไมทัน จึงตองเก็บคาที่เกินไว
ในหนวยความจํ าสํ าหรับแปนพิมพ (KEYBOARD BUFFER)
โดยคาที่เปนไปไดอยูระหวาง 0 ถึง 4096
: ตัวอยางที่ 2.107
        SET TYPEAHEAD TO 128
        CLS                     // เกบ็คาลงหนวยความจํ าของแปนพิมพ
        KEYBOARD "ABCDEF"+CHR(13)+CHR(13)+CHR(65)+CHR(66)
        ? NEXTKEY(),INKEY()     //      97      97
        ? NEXTKEY(),INKEY()     //      98      98
        ? NEXTKEY(),INKEY()     //      99      99
        ? NEXTKEY(),NEXTKEY()   //     100     100
        WAIT ""                 // D
        ACCEPT TO TESTKEY1      // EF
        INKEY(0)                // A
        WAIT ""                 //
        ? NEXTKEY(),LASTKEY()   //      66      65

&  2.91 SET WRAP
! โครงสรางไวยากรณ
SET WRAP ON | OFF | <ตรรกนิพจน>
P วัตถุประสงค
ค ําสัง่นี ้อนญุาติใหการใชคํ าสั่ง PROMPT สะดวกขึ้น เพราะเมื่อเลือก
ตวัเลอืกไปถึงตัวบนสุดแลว กดลูกศรขึ้นสามารถคุมใหอยูกับที่ หรือให
ไปยงัตวัลางสุด ในทางกลับกันเมื่ออยูที่ตัวลางสุด กดลูกศรลงอีกครั้ง
จะมาชีย้งัตัวแรก หรือยังคงอยูที่ตัวลางสุดได
โดย ON หมายถึง อนุญาติใหมองเห็นตัวเลือกตัวแรก และสุดทายตอกัน
และ OFF หมายถึง ไมใหตัวเลือกแรก และสุดทายตอกัน

: ตัวอยางที่ 2.108
        SET MESSAGE TO 22 CENTRE
        SET WRAP ON
        MSS = "แปลวา "
        @ 6,5 PROMPT "ANT"    MESSAGE MSS+"มด"
        @ 7,5 PROMPT "BAT"    MESSAGE MSS+"คางคาว"
        @ 8,5 PROMPT "CAT"    MESSAGE MSS+"แมว"
        @ 9,5 PROMPT "DOG"   MESSAGE MSS+"สุนัข"
        MENU TO OPT
        DO CASE
           CASE OPT = 0
                QUIT
           CASE OPT > 0
                ? "O.K."
        ENDCASE

&  2.92 SKIP
! โครงสรางไวยากรณ
SKIP [<จ ํานวนเรคอรด>] [ALIAS <สมนาม> | <เลขระบุพื้นที่ท ํางาน>]
P วัตถุประสงค
ค ําสั่งนี้ เลื่อนตัวชี้ไปเรคอรดตอไป
: ตัวอยางที่ 2.109
        USE FILEA
        DO WHILE .NOT. EOF()
           ? FIELD1,FIELD2,FIELD3
           SKIP
        ENDDO
: ตัวอยางที่ 2.110
        USE FILEA
        GO BOTTOM
        DO WHILE .NOT. BOF()
           ? FIELD1,FIELD2,FIELD3
           SKIP -1
        ENDDO

&  2.93 SORT
! โครงสรางไวยากรณ
SORT TO <ชือ่แฟม> ON <ชื่อฟลด> [/[A | D][C]]
   [, <ชือ่ฟลด> [/[A | D][C]...]
   [<ชวงทีต่องการ>] [WHILE <เงื่อนไข>] [FOR <เงื่อนไข>]
P วัตถุประสงค
ค ําสัง่นี ้จัดเรียงตามฟลดแลวเก็บลงแฟมใหม
: ตัวอยางที่ 2.111
        USE FILEA
        SORT ON FIELD2 TO SFILEA
        CLOSE ALL
        USE SFILEA
        LIST FIELD1,FIELD2,FIELD3

&  2.94 STORE
! โครงสรางไวยากรณ
STORE <นพิจน> TO <รายการตัวแปร>
<ตวัแปร> = <นิพจน>
<ตวัแปร> := [ <ตัวแปรที่ 2 := ...] <นิพจน>
P วัตถุประสงค
ค ําสัง่นี ้จัดเก็บคาลงในตัวแปร
: ตัวอยางที่ 2.112
        STORE  5 TO A,B,C
        X := Y := Z := 'QC'
        AX1 = 2
        STORE AX1 TO AX2
        ? A,Z,B,Y,C,X
        ? AX1,AX2

&  2.95 SUM
! โครงสรางไวยากรณ
SUM <ชือ่ฟลด> TO <รายการตัวแปร>
   [<ชวงทีต่องการ>] [WHILE <เงื่อนไข>] [FOR <เงื่อนไข>]

P วัตถุประสงค
ค ําสั่งนี้ หาผลรวมของฟลด
: ตัวอยางที่ 2.113
        USE FILEA
        COUNT TO CNT
        SUM FIELD2 TO X1
        SUM FIELD2,FIELD3 TO SUM1,SUM2
        ? SUM1/CNT                      // 2271.25
        ? SUM2/CNT                      //  367.50

&  2.96 TEXT
! โครงสรางไวยากรณ
TEXT [TO PRINTER] [TO FILE <ชือ่แฟม>]
     <ขอความ>...
ENDTEXT
P วัตถุประสงค
ค ําสัง่นี ้แสดงขอความออกทางอุปกรณแสดงผล
: ตัวอยางที่ 2.114
        TEXT
           THIS IS THE TEST OF TEXT COMMAND.
           =================================
           ABCD
           EFGH
           IJKL        // ขอความทั้ง 5 บรรทัดจะแสดงออกทางจอภาพ
        ENDTEXT
: ตัวอยางที่ 2.115
        XDATE := DTOC(DATE())
        TEXT TO PRINTER
           TODAY IS &XDATE
           HAPPY BIRTHDAY TO YOU
        ENDTEXT
        EJECT

&  2.97 TOTAL
! โครงสรางไวยากรณ
TOTAL ON <ชือ่ฟลด> [FIELDS <รายการชื่อฟลด>] TO <ชื่อแฟม>
   [<ชวงทีต่องการ>] [WHILE <เงื่อนไข>] [FOR <เงื่อนไข>]
P วัตถุประสงค
ค ําสัง่นี ้หาผลรวม โดยรวมคาเฉพาะที่เปนกลุมเดียวกัน เก็บลงแฟมใหม ในฟลดเดิม
: ตัวอยางที่ 2.116
        USE FILEF INDEX IFIELD1
        // จบักลุมตาม FIELD1 โดยหาผลรวมของ FIELD2
        TOTAL ON &(INDEXKEY(0)) FIELDS FIELD2 TO TFILEF1
        // จบักลุมตาม FIELD1 โดยหาผลรวมของ FIELD2 และ FIELD3
        TOTAL ON &(INDEXKEY(0)) FIELDS FIELD2,FIELD3 TO TFILEF2

&  2.98 TYPE
! โครงสรางไวยากรณ
TYPE <ชือ่แฟม> [TO PRINTER] [TO FILE <ชื่อแฟมผลลัพธ>]
P วัตถุประสงค
ค ําสัง่นี ้แสดงแฟมตัวอักษรออกทางอุปกรณแสดงผล
: ตัวอยางที่ 2.117
        TYPE X.PRG
        INKEY(0)
        TYPE X.PRG TO PRINTER

&  2.99 UNLOCK
! โครงสรางไวยากรณ
UNLOCK [ALL]
P วัตถุประสงค
ค ําสั่งนี ้ยกเลิกการจองแฟม
: ตัวอยางที่ 2.118
        USE FILEA
        USE FILEA NEW
        SELE 1
        UNLOCK
        CLOSE

        SELE 2
        USE FILEA NEW
        DELETE
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)

&  2.100 UPDATE
! โครงสรางไวยากรณ
UPDATE FROM <สมนาม>
   ON <ชือ่ฟลด> [RANDOM]
   REPLACE <ชือ่ฟลด> WITH <ชื่อตัวแปร>
           [,<ชือ่ฟลดที ่2> WITH <ชื่อตัวแปรที่ 2>...]
P วัตถุประสงค
ค ําสัง่นี ้ปรับปรุงแฟมขอมูล โดยใชอีกแฟมหนึ่งมาแกไข
: ตัวอยางที่ 2.119
        USE FILEU       // มโีครงสรางเหมือนแฟม FILEA
        USE FILEA INDEX IFIELD1 NEW    ;        REINDEX
        UPDATE FROM FILEU ON FIELD1 RANDOM;
               REPLACE FIELD2 WITH FIELD2 + FILEU->FIELD2,;
                       FIELD3 WITH FIELD3 + FILEU->FIELD3

&  2.101 USE
! โครงสรางไวยากรณ
USE  [<ชือ่แฟม>
     [INDEX <รายการชื่อแฟมดรรชนี>]
     [ALIAS <สมนาม>] [EXCLUSIVE | SHARED]
     [NEW] [READONLY]
P วัตถุประสงค
ค ําสั่งนี ้เปดแฟมมาประมวลผล
: ตัวอยางที่ 2.220
        USE FILEA READONLY
        // DELETE   ใชค ําสั่งนี้ไมได เพราะแฟมถูกอนุญาติใหอานเทานั้น
        LIST FIELD1,FIELD2,FIELD3
        ? RECCOUNT(),RECNO()       //  4     5

&  2.102 WAIT
! โครงสรางไวยากรณ
WAIT [<ขอความ>] [TO <ชื่อตัวแปร>]
P วัตถุประสงค
ค ําสั่งนี ้หยุดรอใหกดปุมใด ๆ
: ตัวอยางที่ 2.221
        USE FILEA
        WHILE .NOT. EOF()
             DISPLAY FIELD1,FIELD2,FIELD3
             WAIT    // แสดงค ําวา PRESS ANY KEY TO CONTINUE...
             SKIP
        END
: ตัวอยางที่ 2.222
        WAIT "PRESS A TO D" TO GETCHR  // แสดงตัวอักษรที่กดใหเห็น
        IF GETCHR $ [ABCDABCD]
           ? "YOU ARE RIGHT."
        ELSE
           ? "YOU ARE WRONG."
        ENDIF

&  2.103 ZAP
! โครงสรางไวยากรณ
ZAP
P วัตถุประสงค
ค ําสัง่นี ้ลบเรคอรดทั้งหมดจากแฟม โดยไมจํ าเปนตองใชคํ าสั่ง PACK
: ตัวอยางที่ 2.223
        USE FILEA
        ZAP
        ? RECCOUNT()    // 0
        CLOSE
        COPY FILE FABAK.DBF TO FILEA.DBF
        USE FILEA.DBF
        LIST FIELD1,FIELD2,FIELD3
        ? RECCOUNT()    // 4

ยิ่งเรียน ยิ่งรูวาไมรู
จึงตองพยายามมากยิ่งขึ้น

เพือ่ลดความไมรูใหมากที่สุด
และน ําความรูไปหาเลี้ยงชีพ .. ตอไป

